

## TP 10 : Corrigé des travaux pratiques surveillés

### Exercice 1 (Théorème du point fixe)

3. Je trouve

```
y1 =
    1.165827
x1 =
    1.1666667
n1 =
    3.
```

Le dernier chiffre dépend de la manière dont vous avez cliqué sur la souris. Une approximation est donc  $x = 1.165$ .

4. Par exemple la fonction définie par

```
function res = iter(f,u0,n)
    res = u0;
    for i=1:n
        res = f(res);
    end;
endfunction
convient.
```

5. La fonction s'écrit

```
function res=f(x)
    res=atan(2*x);
endfunction
```

6. Après avoir effectué

```
-->getf f.sci
```

```
-->getf iter.sci
```

On tape successivement

```
-->u5=iter(f,1,5)
```

```
u5 =
    1.1649806
```

```
-->u10=iter(f,1,10)
```

```
u10 =
    1.1655595
```

```
-->u20=iter(f,1,20)
```

```
u20 =
    1.1655612
```

```
-->f(u20)-u20
```

```
ans =
    9.785D-12
```

On constate que  $u_{20}$  est solution de l'équation  $f(x) = x$  à  $10^{-11}$  près.

7. La modification de la procédure de dichotomie donne

```
function res=ndicho(f,a,b,n)
  for i=1:n
    m = (a+b)/2;
    if (f(m)*f(a)<0) then b=m; else a=m; end;
  end
  res =(a+b)/2;
endfunction
```

8. La fonction  $x - f(x)$  s'implémente comme suit :

```
function res=xmf(x)
  res=x-atan(2*x);
endfunction
```

9. On tape successivement :

```
-->getf ndicho.sci

-->getf xmf.sci

-->v20=ndicho(xmf,1,2,20)
v20 =

    1.1655612

-->f(v20)-v20
ans =

   - 9.635D-09
```

10. On constate qu'à nombre d'itération égales, la première méthode donne un résultat plus précis que la méthode de dichotomie.

### Exercice 2 (Méthode de la sécante)

0. La droite  $\Delta_u$  a pour coefficient directeur

$$\frac{f(b) - f(u)}{b - u}$$

et passe en  $y = f(u)$  en  $x = u$ . L'équation de la droite est donc

$$y = \frac{f(b) - f(u)}{b - u}(x - u) + f(u)$$

Elle coupe l'axe des  $x$  lorsque  $y = 0$  ce qui donne

$$x_1 = u - \frac{b - u}{f(b) - f(u)} f(u)$$

1. La fonction s'écrit donc

```
function res=secante(g,a,b,n)
  u=a
  for i=1:n
    y=g(u);
    u=u-(b-u)/(g(b)-y)*y;
  end
  res=u;
endfunction;
```

2. On tape successivement

```
-->getf secante.sci

-->w15=secante(xmf,1,2,15)
```

```
w15 =
```

```
1.1655612
```

```
-->w20=secante(xmf,1,2,20)
```

```
w20 =
```

```
1.1655612
```

3. *Les évaluations demandées s'écrivent*

```
-->f(w15)-w15
```

```
ans =
```

```
4.463D-14
```

```
-->f(w20)-w20
```

```
ans =
```

```
0.
```

*On constate que la méthode est supérieure à la méthode itérative et a plus forte raison à la méthode de dichotomie car en seulement 15 itérations elle offre une précision meilleure. En 20 itérations l'erreur est en dessous de la précision de Scilab (qui est de l'ordre de  $10^{-16}$ ).*